## Title

irtgraph icc — Item characteristic curve plot

Description	Quick start	Menu	Syntax
Options	Remarks and examples	Stored results	Also see

## Description

irtgraph icc plots item characteristic curves (ICCs) for binary items and category characteristic curves (CCCs) for categorical items for the currently fitted IRT model.

## **Quick start**

1PL model for binary items b1 to b10

irt 1pl b1-b10

Plot ICCs for all items

irtgraph icc

Plot ICCs and item difficulties for items b1, b5, and b9

irtgraph icc b1 b5 b9, blocation

GRM for ordinal items o1 to o5, items coded 1, 2, 3

irt grm o1-o5

Plot CCCs for selected item categories

irtgraph icc 1.01 3.05 2.01

Plot CCCs for the first category of all items

irtgraph icc 1.o\*

### Menu

Statistics > IRT (item response theory)

## Syntax

```
Basic syntax
irtgraph icc [varlist] [, options]
```

Full syntax

ccc

line\_options

```
irtgraph icc (varlist [, plot_options]) (varlist [, plot_options]) [...] [, options]
```

varlist is a list of items from the currently fitted IRT model.

options	Description
Plots	
<u>blo</u> cation[( <i>line_options</i> )]	add vertical lines for estimated item difficulties
<pre>plocation[(line_options)] bcc ccc range(# #)</pre>	add horizontal lines for midpoint probabilities plot boundary characteristic curves for categorical items plot category characteristic curves plot over $\theta = \#$ to $\#$
Line line_options	affect rendition of the plotted curves
Add plots addplot( <i>plot</i> )	add other plots to the ICC plot
Y axis, X axis, Titles, Legend, Overall twoway_options	any options other than by() documented in [G-3] twoway_options
Data n(#) data( <i>filename</i> [, replace])	evaluate curves at # points; default is n(300) save plot data to a file
plot_options	Description
<pre>blocation[(line_options)]</pre>	add vertical lines for estimated item difficulties
<pre>plocation[(line_options)] bcc</pre>	add horizontal lines for midpoint probabilities plot boundary characteristic curves for categorical items

plot category characteristic curves

affect rendition of the plotted curves

*varlist* may use factor-variable notation; see [U] **11.4.3 Factor variables**. *line\_options* in *plot\_options* override the same options specified in *options*.

## Options

Plots

- $blocation[(line_options)]$  specifies that for each ICC, a vertical line be drawn from the estimated difficulty parameter on the x axis to the curve. The optional *line\_options* specify how the vertical lines are rendered; see [G-3] *line\_options*. This option implies option bcc.
- plocation  $|(line_options)|$  specifies that for each ICC, a horizontal line be drawn from the midpoint probability on the y axis to the curve. The optional *line\_options* specify how the horizontal lines are rendered; see [G-3] *line\_options*. This option implies option bcc.
- bcc specifies that boundary characteristic curves (BCCs) be plotted for categorical items. The ICCs for the individual item categories are plotted by default. This option has no effect on binary items.
- ccc specifies that category characteristic curves (CCCs) be plotted for all items. This is the default behavior for categorical items. For binary items, this option will plot ICCs for both outcomes.
- range (# #) specifies the range of values for  $\theta$ . This option requires a pair of numbers identifying the minimum and maximum. The default is range (-4 4) unless the estimated difficulty parameters exceed these values, in which case the range is extended.

Line

*line\_options* affect the rendition of the plotted ICCs; see [G-3] *line\_options*.

Add plots

addplot(plot) allows adding more graph twoway plots to the graph; see [G-3] addplot\_option.

🛛 Y axis, X axis, Titles, Legend, Overall 🗋

*twoway\_options* are any of the options documented in [G-3] *twoway\_options*, excluding by(). These include options for titling the graph (see [G-3] *title\_options*) and for saving the graph to disk (see [G-3] *saving\_option*).

Data

n(#) specifies the number of points at which the ICCs, CCCs, and BCCs are to be evaluated. The default is n(300).

data(*filename* |, replace |) saves the plot data to a Stata data file.

### **Remarks and examples**

#### stata.com

irtgraph icc plots ICCs for binary items and CCCs for categorical items after estimating the parameters of an IRT model using irt.

ICCs are also known as item response functions and item response curves.

CCCs are also known as category response functions, option response functions, operating characteristic curves, and category response curves.

For categorical items, irtgraph icc also plots BCCs, which are probability curves for crossing a boundary. BCCs are also known as "category boundary curves".

irtgraph icc is very flexible, and the best way to learn its capabilities is through examples.

### Example 1: ICCs for binary outcomes

We continue with the model from example 1 of [IRT] irt 1pl. Recall that we fit a 1PL model to the nine binary items. Here we use estat report to rearrange the estimated IRT parameters sorted by item difficulty.

```
. use http://www.stata-press.com/data/r14/masc1
(Data from De Boeck & Wilson (2004))
. irt 1pl q1-q9
Fitting fixed-effects model:
Iteration 0:
               \log likelihood = -4275.6606
Iteration 1:
               \log likelihood = -4269.7861
Iteration 2:
               log likelihood = -4269.7825
Iteration 3:
             log likelihood = -4269.7825
Fitting full model:
Iteration 0:
               log likelihood = -4153.3609
               log likelihood = -4142.374
Iteration 1:
               log likelihood = -4142.3516
Iteration 2:
               log likelihood = -4142.3516
Iteration 3:
One-parameter logistic model
Log likelihood = -4142.3516
```

Number of obs = 800

		Coef.	Std. Err.	z	P> z	[95% Conf.	Interval]
	Discrim	.852123	.0458445	18.59	0.000	.7622695	.9419765
q1	Diff	7071339	.1034574	-6.84	0.000	9099066	5043612
q2	Diff	1222008	.0963349	-1.27	0.205	3110138	.0666122
q3	Diff	-1.817693	.1399523	-12.99	0.000	-2.091994	-1.543391
q4	Diff	.3209596	.0976599	3.29	0.001	.1295498	.5123695
q5	Diff	1.652719	.1329494	12.43	0.000	1.392144	1.913295
q6	Diff	.6930617	.1031842	6.72	0.000	.4908243	.8952991
q7	Diff	1.325001	.1205805	10.99	0.000	1.088668	1.561335
q8	Diff	-2.413443	.1691832	-14.27	0.000	-2.745036	-2.08185
q9	Diff	-1.193206	.1162054	-10.27	0.000	-1.420965	965448

. estat report	t, sort(b) by						
One-parameter logistic model Log likelihood = -4142.3516				Number	800		
	Coef.	Std. Err.	Z	P> z	[95% Conf.	Interval]	
Discrim	.852123	.0458445	18.59	0.000	.7622695	.9419765	
Diff							
q8	-2.413443	.1691832	-14.27	0.000	-2.745036	-2.08185	
q3	-1.817693	.1399523	-12.99	0.000	-2.091994	-1.543391 965448	
q9	-1.193206	.1162054	-10.27	0.000	-1.420965		
q1	7071339	.1034574	-6.84	0.000	9099066	5043612	
q2	1222008	.0963349	-1.27	0.205	3110138	.0666122	
q4	.3209596	.0976599	3.29	0.001	.1295498	.5123695	
q6	.6930617	.1031842	6.72	0.000	.4908243	.8952991	
q7	1.325001	.1205805	10.99	0.000	1.088668	1.561335	
q5	1.652719	.1329494	12.43	0.000	1.392144	1.913295	

To plot ICCs for all items in the model, we simply type

. irtgraph icc



With many items, the default graph is not pretty. The legend takes up too much space. One way to make the graph more readable is to shrink the legend and move it inside the plot region. We also specify a list of items explicitly so that the legend lists the ICCs in the order they appear in the graph. At the end of our interactive session, we came up with the following.

- . irtgraph icc q8 q3 q9 q1 q2 q4 q6 q7 q5,  $\,$
- > legend(pos(4) col(1) ring(0) size(small))



Another way to present the ICCs is to turn off the legend and highlight the items with the lowest and highest estimated difficulty parameter. From the output of estat report, we see that item q8 has the lowest estimated difficulty parameter and item q5 has the highest estimated difficulty parameter. Here we put those items in a separate plotting group, request their estimated difficulty locations be plotted, and put the remaining items in another plotting group.

. irtgraph icc (q5 q8, blocation) (q1-q4 q6 q7 q9), legend(off)



This plot shows the information we wanted, but we can tinker further to make items q5 and q8 stand out and make the whole plot more dramatic. Here is the final result.

```
. irtgraph icc
> (q8 q5, lcolor(black) lwidth(thick) bloc(lcolor(black)))
> (q1-q4 q6 q7 q9, lpattern(dash)),
> range(-5 5) xlabel(-5 -2.41 0 1.65 5) legend(off) lcolor(red)
```



We admit the above works nicely for a 1PL model because the ICCs do not cross; 2PL and 3PL models may require a different approach, but the general idea remains the same—we rarely obtain the desired ICC plot on the first try and need to work incrementally to arrive at the graph that best suits the estimated model parameters.

4

#### Technical note

For a binary item, it is standard practice to plot only the ICC for the probability of the positive outcome. Thus the following commands are equivalent.

. irtgraph icc q1 . irtgraph icc 1.q1

However, there are in fact two ICCs we could plot: one for the probability of the positive outcome and one for the probability of the negative outcome. To plot both ICCs, we can use any of the following:

```
. irtgraph icc 0.q1 1.q1
. irtgraph icc i.q1
. irtgraph icc q1, ccc
```

Because the two probabilities sum to 1, the ICC for the negative outcome is a mirror image of the ICC for the positive outcome reflected about the y axis at 0.5.

#### Example 2: CCCs for categorical outcomes

We continue with the model introduced in example 1 of [IRT] irt grm. To easily present some graphical features, we collapse the last two categories into one for all items and refit the GRM.

```
. use http://www.stata-press.com/data/r14/charity
(Data from Zheng & Rabe-Hesketh (2007))
. recode ta1-ta5 (3=2)
(ta1: 58 changes made)
(ta2: 102 changes made)
(ta3: 55 changes made)
(ta4: 36 changes made)
(ta5: 86 changes made)
. irt grm ta1-ta5
Fitting fixed-effects model:
Iteration 0:
                log likelihood = -4713.2513
Iteration 1:
                \log likelihood = -4713.2513
Fitting full model:
Iteration 0:
                \log likelihood = -4519.7174
                \log likelihood = -4434.2749
Iteration 1:
Iteration 2:
                \log likelihood = -4430.3549
Iteration 3:
                log likelihood =
                                   -4430.343
Iteration 4:
                log likelihood = -4430.343
Graded response model
                                                   Number of obs
                                                                                945
Log likelihood =
                   -4430.343
                              Std. Err.
                     Coef.
                                              7.
                                                   P>|z|
                                                              [95% Conf. Interval]
ta1
     Discrim
                  .8184045
                              .0951662
                                            8.60
                                                   0.000
                                                              .6318823
                                                                           1.004927
        Diff
        >=1
                 -1.668228
                              .1931402
                                           -8.64
                                                   0.000
                                                             -2.046775
                                                                           -1.28968
          =2
                  1.423835
                              .1696051
                                            8.39
                                                   0.000
                                                              1.091415
                                                                           1.756255
ta2
                  1.031415
                              .1119233
                                            9.22
                                                   0.000
                                                              .8120498
                                                                           1.250781
     Discrim
        Diff
        >=1
                 -1.563834
                              .1578408
                                           -9.91
                                                   0.000
                                                             -1.873196
                                                                          -1.254472
         =2
                  -.007408
                               .077598
                                           -0.10
                                                   0.924
                                                             -.1594973
                                                                           .1446814
ta3
     Discrim
                  1.595058
                              .1519538
                                           10.50
                                                   0.000
                                                              1.297234
                                                                           1.892882
        Diff
        >=1
                 -1.128531
                                .09129
                                          -12.36
                                                   0.000
                                                             -1.307457
                                                                          -.9496062
         =2
                  1.082083
                                           12.16
                                                   0.000
                                                              .9077368
                                                                           1.256429
                              .0889538
ta4
                                            9.87
     Discrim
                  1.814543
                              .1837944
                                                   0.000
                                                              1.454313
                                                                           2.174773
        Diff
                                          -5.81
                                                   0.000
        >=1
                 -.3493225
                              .0601007
                                                             -.4671178
                                                                          -.2315273
         =2
                  1.527393
                              .1080719
                                           14.13
                                                   0.000
                                                              1.315576
                                                                            1.73921
ta5
     Discrim
                  1.528601
                              .1497992
                                           10.20
                                                   0.000
                                                                           1.822202
                                                                 1.235
        Diff
        >=1
                  -.825658
                              .0811847
                                          -10.17
                                                   0.000
                                                             -.9847772
                                                                          -.6665389
         =2
                  .6543883
                              .0739536
                                            8.85
                                                   0.000
                                                               .509442
                                                                           .7993347
```

For a model with many categorical items, we do not recommend using irtgraph icc without *varlist*, because the resulting graph will contain far too many plotted curves. With 5 items, each with 3 categories, the total number of CCCs in the default plot is 15. Here we focus on item ta1.

```
. irtgraph icc ta1, xlabel(-4 -1.45 1.20 4, grid)
```



In a GRM, the adjacent probability curves do not cross at the estimated difficulty parameters. Each crossing point represents the level of the latent trait at which an examinee "transitions" from responding in one category versus the next. Thus, in the graph above, respondents whose trait level is below approximately -1.45 are most likely to answer 0 (strongly agree); respondents whose trait level is between approximately -1.45 and 1.20 are most likely to answer 1 (somewhat agree); and respondents whose trait level is above approximately 1.20 are most likely to answer 2 (somewhat or strongly disagree).

Because the GRM is defined in terms of cumulative probabilities, the estimated difficulties represent a point at which a person with  $\theta = b_{ik}$  has a 50% chance of responding in category k or higher. We can use irtgraph to plot these probabilities with the corresponding estimated category difficulties. These probability curves are known as BCCs. We specify option blocation, which plots the category difficulties and also implies option bcc.

. irtgraph icc ta1, blocation



Technical note

In the example above, we typed

. irtgraph icc ta1

to plot the CCCs for item ta1. Because item ta1 is coded 0, 1, or 2, we could have typed

. irtgraph icc 0.ta1 1.ta1 2.ta1

or

. irtgraph icc i.ta1

However, the first notation is most convenient to type. The factor notation comes in handy when we want to plot a particular category or change its appearance in the graph.

### Example 3: Combining graphs

Sometimes, it is useful to focus on one category and plot its CCCs for all items. Below we show one way of presenting this information. We do not label the curves, because we want to see only the overall shape and location of the CCCs. We could always play with the legend to identify the individual curves, as we did above in example 1.

```
. irtgraph icc 0.ta*, legend(off) title("") lcolor(black) nodraw
> name(out0,replace)
. irtgraph icc 1.ta*, legend(off) title("") lcolor(red) nodraw
> name(out1,replace)
. irtgraph icc 2.ta*, legend(off) title("") lcolor(blue) nodraw
> name(out2,replace)
. graph combine out0 out1 out2, col(1) xsize(3) ysize(6)
> title("CCCs for items ta1-ta5")
```



# **Stored results**

irtgraph icc stores the following in r():

Macros r(xvals)

r(xvals)	values	used	to	label	the	x	axis
r(yvals)	values	used	to	label	the	y	axis

٩

### Also see

- [IRT] irt Introduction to IRT models
- [IRT] irt 1pl One-parameter logistic model
- [IRT] irt 2pl Two-parameter logistic model
- [IRT] irt 3pl Three-parameter logistic model
- [IRT] irt grm Graded response model
- [IRT] irt hybrid Hybrid IRT models
- [IRT] irt nrm Nominal response model
- [IRT] irt pcm Partial credit model
- [IRT] irt rsm Rating scale model
- [IRT] irtgraph tcc Test characteristic curve plot